

# Stirling's formula

Manuel Eberl

March 16, 2017

## Abstract

This work contains a proof of Stirling's formula both for the factorial  $n! \sim \sqrt{2\pi n}(n/e)^n$  on natural numbers and the real Gamma function  $\Gamma(x) \sim \sqrt{2\pi/x}(x/e)^x$ . The proof is based on work by Graham Jameson [1].

## Contents

**theory** *Stirling-Formula*

**imports**

~~/src/HOL/Analysis

../Landau-Symbols/Landau-Symbols

**begin**

**context**

**begin**

First, we define the  $S_n^*$  from Jameson's article:

**private definition**  $S' :: \text{nat} \Rightarrow \text{real} \Rightarrow \text{real}$  **where**

$$S' n x = 1/(2*x) + (\sum_{r=1..<n.} 1/(of-nat r+x)) + 1/(2*(n+x))$$

Next, the trapezium (also called  $T$  in Jameson's article):

**private definition**  $T :: \text{real} \Rightarrow \text{real}$  **where**

$$T x = 1/(2*x) + 1/(2*(x+1))$$

Now we define The difference  $\Delta(x)$ :

**private definition**  $D :: \text{real} \Rightarrow \text{real}$  **where**

$$D x = T x - \ln(x+1) + \ln x$$

**private lemma** *S'-telescope-trapezium*:

**assumes**  $n > 0$

**shows**  $S' n x = (\sum_{r<n.} T (of-nat r + x))$

**proof** (*cases n*)

**case** (*Suc m*)

**hence**  $m: \text{Suc } m = n$  **by** *simp*

**have**  $(\sum r < n. T (of\text{-}nat\ r + x)) =$   
 $(\sum r < Suc\ m. 1 / (2 * real\ r + 2 * x)) + (\sum r < n. 1 / (2 * real\ (Suc\ r)$   
 $+ 2 * x))$   
**unfolding**  $m$  **by**  $(simp\ add: T\text{-}def\ sum.\ distrib\ algebra\ \text{simps})$   
**also have**  $(\sum r < Suc\ m. 1 / (2 * real\ r + 2 * x)) =$   
 $1 / (2 * x) + (\sum r < m. 1 / (2 * real\ (Suc\ r) + 2 * x))$  **(is - = ?a + ?S)**  
**by**  $(subst\ sum\ \text{lessThan}\ \text{Suc}\ \text{shift})\ simp$   
**also have**  $(\sum r < n. 1 / (2 * real\ (Suc\ r) + 2 * x)) =$   
 $?S + 1 / (2 * (real\ m + x + 1))$  **(is - = - + ?b)** **by**  $(simp\ add: Suc)$   
**also have**  $?a + ?S + (?S + ?b) = 2 * ?S + ?a + ?b$  **by**  $(simp\ add: add\ \text{ac})$   
**also have**  $2 * ?S = (\sum r = 0 .. < m. 1 / (real\ (Suc\ r) + x))$   
**unfolding**  $sum\ \text{distrib}\ \text{left}$  **by**  $(intro\ sum.\ \text{cong})\ (auto\ simp\ add: divide\ \text{simps})$   
**also have**  $(\sum r = 0 .. < m. 1 / (real\ (Suc\ r) + x)) = (\sum r = Suc\ 0 .. < Suc\ m. 1 /$   
 $(real\ r + x))$   
**by**  $(subst\ sum.\ atLeast\ \text{Suc}\ \text{lessThan}\ \text{Suc}\ \text{shift})\ simp\ \text{all}$   
**also have**  $\dots = (\sum r = 1 .. < n. 1 / (real\ r + x))$  **unfolding**  $m$  **by**  $simp$   
**also have**  $\dots + ?a + ?b = S'\ n\ x$  **by**  $(simp\ add: S'\ \text{def}\ Suc)$   
**finally show**  $?thesis\ ..$   
**qed**  $(insert\ \text{assms},\ simp\ \text{all})$

**private lemma** *stirling-trapezium:*

**assumes**  $x: (x :: real) > 0$   
**shows**  $D\ x \in \{0 .. 1 / (12 * x^2) - 1 / (12 * (x + 1)^2)\}$   
**proof** -  
**define**  $y$  **where**  $y = 1 / (2 * x + 1)$   
**from**  $x$  **have**  $y: y > 0\ y < 1$  **by**  $(simp\ \text{all}\ add: divide\ \text{simps}\ y\ \text{def})$   
  
**from**  $x$  **have**  $D\ x = T\ x - \ln\ ((x + 1) / x)$  **by**  $(subst\ \ln\ \text{div})\ (simp\ \text{all}\ add: D\ \text{def})$   
**also from**  $x$  **have**  $(x + 1) / x = 1 + 1 / x$  **by**  $(simp\ add: field\ \text{simps})$   
**finally have**  $D: D\ x = T\ x - \ln\ (1 + 1 / x)$  .  
  
**from**  $y$  **have**  $(\lambda n. y * y^n)\ sums\ (y * (1 / (1 - y)))$   
**by**  $(intro\ geometric\ \text{sums}\ sums\ \text{mult})\ simp\ \text{all}$   
**hence**  $(\lambda n. y ^ Suc\ n)\ sums\ (y / (1 - y))$  **by**  $simp$   
**also from**  $x$  **have**  $y / (1 - y) = 1 / (2 * x)$  **by**  $(simp\ add: y\ \text{def}\ divide\ \text{simps})$   
**finally have**  $*$ :  $(\lambda n. y ^ Suc\ n)\ sums\ (1 / (2 * x))$  .  
  
**from**  $y$  **have**  $(\lambda n. (-y) * (-y)^n)\ sums\ ((-y) * (1 / (1 - (-y))))$   
**by**  $(intro\ geometric\ \text{sums}\ sums\ \text{mult})\ simp\ \text{all}$   
**hence**  $(\lambda n. (-y) ^ Suc\ n)\ sums\ (-y / (1 + y))$  **by**  $simp$   
**also from**  $x$  **have**  $y / (1 + y) = 1 / (2 * (x + 1))$  **by**  $(simp\ add: y\ \text{def}\ divide\ \text{simps})$   
**finally have**  $**$ :  $(\lambda n. (-y) ^ Suc\ n)\ sums\ (-1 / (2 * (x + 1)))$  .  
  
**from**  $sums\ \text{diff}[OF\ *\ **]$  **have**  $sum1: (\lambda n. y ^ Suc\ n - (-y) ^ Suc\ n)\ sums\ T\ x$   
**by**  $(simp\ add: T\ \text{def})$   
  
**from**  $y$  **have**  $abs\ y < 1\ abs\ (-y) < 1$  **by**  $simp\ \text{all}$   
**from**  $sums\ \text{diff}[OF\ this[THEN\ \ln\ \text{series}]]$

**have**  $(\lambda n. y^n / \text{real } n - (-y)^n / \text{real } n) \text{ sums } (\ln (1 + y) - \ln (1 - y))$  **by** *simp*  
**also from**  $y$  **have**  $\ln (1 + y) - \ln (1 - y) = \ln ((1 + y) / (1 - y))$  **by** (*simp add: ln-div*)  
**also from**  $x$  **have**  $(1 + y) / (1 - y) = 1 + 1/x$  **by** (*simp add: divide-simps y-def*)  
**finally have**  $(\lambda n. y^n / \text{real } n - (-y)^n / \text{real } n) \text{ sums } \ln (1 + 1/x)$  .  
**hence**  $\text{sum2}: (\lambda n. y^{\text{Suc } n} / \text{real } (\text{Suc } n) - (-y)^{\text{Suc } n} / \text{real } (\text{Suc } n)) \text{ sums } \ln (1 + 1/x)$   
**by** (*subst sums-Suc-iff*) *simp*

**from**  $\text{sum2 sum1}$  **have**  $\ln (1 + 1/x) \leq T x$   
**proof** (*rule sums-le [OF allI, rotated]*)  
**fix**  $n :: \text{nat}$   
**show**  $y^{\text{Suc } n} / \text{real } (\text{Suc } n) - (-y)^{\text{Suc } n} / \text{real } (\text{Suc } n) \leq y^{\text{Suc } n} - (-y)^{\text{Suc } n}$   
**proof** (*cases even n*)  
**case** *True*  
**hence**  $\text{eq}: A - (-y)^{\text{Suc } n} / B = A + y^{\text{Suc } n} / B$   $A - (-y)^{\text{Suc } n} = A + y^{\text{Suc } n}$   
**for**  $A B$  **by** *simp-all*  
**from**  $y$  **show** *?thesis unfolding eq*  
**by** (*intro add-mono*) (*auto simp: divide-simps*)  
**qed** *simp-all*  
**qed**  
**hence**  $D x \geq 0$  **by** (*simp add: D*)

**define**  $c$  **where**  $c = (\lambda n. \text{if even } n \text{ then } 2 * (1 - 1 / \text{real } (\text{Suc } n)) \text{ else } 0)$   
**note** *sums-diff [OF sum1 sum2]*  
**also have**  $(\lambda n. y^{\text{Suc } n} - (-y)^{\text{Suc } n} - (y^{\text{Suc } n} / \text{real } (\text{Suc } n) - (-y)^{\text{Suc } n} / \text{real } (\text{Suc } n))) = (\lambda n. c n * y^{\text{Suc } n})$   
**by** (*intro ext*) (*simp add: c-def algebra-simps*)  
**finally have**  $\text{sum3}: (\lambda n. c n * y^{\text{Suc } n}) \text{ sums } D x$  **by** (*simp add: D*)

**from**  $y$  **have**  $(\lambda n. y^2 * (\text{of-nat } (\text{Suc } n) * y^n)) \text{ sums } (y^2 * (1 / (1 - y)^2))$   
**by** (*intro sums-mult geometric-deriv-sums*) *simp-all*  
**hence**  $(\lambda n. \text{of-nat } (\text{Suc } n) * y^{(n+2)}) \text{ sums } (y^2 / (1 - y)^2)$   
**by** (*simp add: algebra-simps power2-eq-square*)  
**also from**  $x$  **have**  $y^2 / (1 - y)^2 = 1 / (4 * x^2)$  **by** (*simp add: y-def divide-simps*)  
**finally have**  $*$ :  $(\lambda n. \text{real } (\text{Suc } n) * y^{\text{Suc } (\text{Suc } n)}) \text{ sums } (1 / (4 * x^2))$  **by** *simp*

**from**  $y$  **have**  $(\lambda n. y^2 * (\text{of-nat } (\text{Suc } n) * (-y)^n)) \text{ sums } (y^2 * (1 / (1 - (-y)^2)))$   
**by** (*intro sums-mult geometric-deriv-sums*) *simp-all*  
**hence**  $(\lambda n. \text{of-nat } (\text{Suc } n) * (-y)^{(n+2)}) \text{ sums } (y^2 / (1 + y)^2)$   
**by** (*simp add: algebra-simps power2-eq-square*)  
**also from**  $x$  **have**  $y^2 / (1 + y)^2 = 1 / (2^2 * (x+1)^2)$

**unfolding** *power-mult-distrib [symmetric]* **by** (*simp add: y-def divide-simps add-ac*)

**finally have** \*\*:  $(\lambda n. \text{real } (\text{Suc } n) * (-y) ^ (\text{Suc } (\text{Suc } n))) \text{ sums } (1 / (4 * (x + 1)^2))$  **by** *simp*

**define** *d* **where**  $d = (\lambda n. \text{if even } n \text{ then } 2 * \text{real } n \text{ else } 0)$

**note** *sums-diff[OF \*\*]*

**also have**  $(\lambda n. \text{real } (\text{Suc } n) * y ^ (\text{Suc } (\text{Suc } n)) - \text{real } (\text{Suc } n) * (-y) ^ (\text{Suc } (\text{Suc } n))) =$

$(\lambda n. d (\text{Suc } n) * y ^ \text{Suc } (\text{Suc } n))$

**by** (*intro ext*) (*simp-all add: d-def*)

**finally have**  $(\lambda n. d n * y ^ \text{Suc } n \text{ sums } (1 / (4 * x^2) - 1 / (4 * (x + 1)^2)))$

**by** (*subst (asm) sums-Suc-iff*) (*simp add: d-def*)

**from** *sums-mult[OF this, of 1/3] x*

**have** *sum4*:  $(\lambda n. d n / 3 * y ^ \text{Suc } n \text{ sums } (1 / (12 * x^2) - 1 / (12 * (x + 1)^2)))$

**by** (*simp add: field-simps*)

**have**  $D x \leq (1 / (12 * x^2) - 1 / (12 * (x + 1)^2))$

**proof** (*intro sums-le [OF - sum3 sum4] allI*)

**fix**  $n :: \text{nat}$

**define**  $c' :: \text{nat} \Rightarrow \text{real}$

**where**  $c' = (\lambda n. \text{if odd } n \vee n = 0 \text{ then } 0 \text{ else if } n = 2 \text{ then } 4/3 \text{ else } 2)$

**show**  $c n * y ^ \text{Suc } n \leq d n / 3 * y ^ \text{Suc } n$

**proof** (*intro mult-right-mono*)

**have**  $c n \leq c' n$  **by** (*simp add: c-def c'-def*)

**also consider**  $n = 0 \mid n = 1 \mid n = 2 \mid n \geq 3$  **by** *force*

**hence**  $c' n \leq d n / 3$  **by** *cases (simp-all add: c'-def d-def)*

**finally show**  $c n \leq d n / 3$  .

**qed** (*insert y, simp*)

**qed**

**with**  $\langle D x \geq 0 \rangle$  **show** *?thesis* **by** *simp*

**qed**

The following functions correspond to the  $p_n(x)$  from the article. The special case  $n = 0$  would not, strictly speaking, be necessary, but it allows some theorems to work even without the precondition  $n \neq 0$ :

**private definition**  $p :: \text{nat} \Rightarrow \text{real} \Rightarrow \text{real}$  **where**

$p n x = (\text{if } n = 0 \text{ then } 1/x \text{ else } (\sum r < n. D (\text{real } r + x)))$

We can write the Digamma function in terms of  $S'$ :

**private lemma** *S'-LIMSEQ-Digamma*:

**assumes**  $x: x \neq 0$

**shows**  $(\lambda n. \ln (\text{real } n) - S' n x - 1/(2*x)) \longrightarrow \text{Digamma } x$

**proof** –

**define**  $c$  **where**  $c = (\lambda n. \ln (\text{real } n) - (\sum r < n. \text{inverse } (x + \text{real } r)))$

**have** *eventually*  $(\lambda n. 1 / (2 * (x + \text{real } n)) = c n - (\ln (\text{real } n) - S' n x - 1/(2*x)))$  *at-top*

```

using eventually-gt-at-top[of 0::nat]
proof eventually-elim
  fix n :: nat
  assume n: n > 0
  have c n - (ln (real n) - S' n x - 1/(2*x)) =
    -(\sum r<n. inverse (real r + x)) + (1/x + (\sum r=1..<n. inverse (real r
+ x))) + 1/(2*(real n + x))
    using x by (simp add: S'-def c-def field-simps)
  also have 1/x + (\sum r=1..<n. inverse (real r + x)) = (\sum r<n. inverse (real
r + x))
    unfolding lessThan-atLeast0 using n
    by (subst (2) sum-head-upt-Suc) (simp-all add: field-simps)
  finally show 1 / (2 * (x + real n)) = c n - (ln (real n) - S' n x - 1/(2*x))
by simp
qed
moreover have (\lambda n. 1 / (2 * (x + real n))) \longrightarrow 0
  by (rule real-tendsto-divide-at-top tendsto-const filterlim-tendsto-pos-mult-at-top
filterlim-tendsto-add-at-top filterlim-real-sequentially | simp)+
ultimately have (\lambda n. c n - (ln (real n) - S' n x - 1/(2*x))) \longrightarrow 0
  by (rule Lim-transform-eventually)
from tendsto-minus[OF this] have (\lambda n. (ln (real n) - S' n x - 1/(2*x)) - c
n) \longrightarrow 0 by simp
moreover from Digamma-LIMSEQ[OF x] have c \longrightarrow Digamma x by (simp
add: c-def)
ultimately show (\lambda n. ln (real n) - S' n x - 1/(2*x)) \longrightarrow Digamma x
  by (rule Lim-transform [rotated])
qed

```

Moreover, we can give an expansion of  $S'$  with the  $p$  as variation terms.

```

private lemma S'-approx:
  S' n x = ln (real n + x) - ln x + p n x
proof (cases n = 0)
  case True
    thus ?thesis by (simp add: p-def S'-def)
  next
    case False
      hence S' n x = (\sum r<n. T (real r + x))
        by (subst S'-telescope-trapezium) simp-all
      also have ... = (\sum r<n. ln (real r + x + 1) - ln (real r + x) + D (real r +
x))
        by (simp add: D-def)
      also have ... = (\sum r<n. ln (real (Suc r) + x) - ln (real r + x)) + p n x
        using False by (simp add: sum.distrib add-ac p-def)
      also have (\sum r<n. ln (real (Suc r) + x) - ln (real r + x)) = ln (real n + x)
- ln x
        by (subst sum-lessThan-telescope) simp-all
      finally show ?thesis .
qed

```

We define the limit of the  $p$  (simply called  $p(x)$  in Jameson's article):

**private definition**  $P :: \text{real} \Rightarrow \text{real}$  **where**  
 $P\ x = (\sum n. D\ (\text{real}\ n + x))$

**private lemma**  $D$ -summable:

**assumes**  $x: x > 0$   
**shows** summable  $(\lambda n. D\ (\text{real}\ n + x))$   
**proof** –  
**have** \*: summable  $(\lambda n. 1 / (12 * (x + \text{real}\ n)^2) - 1 / (12 * (x + \text{real}\ (\text{Suc}\ n))^2))$   
**by** (rule telescope-summable' real-tendsto-divide-at-top tendsto-const  
filterlim-tendsto-pos-mult-at-top filterlim-pow-at-top  
filterlim-tendsto-add-at-top filterlim-real-sequentially | simp)+  
**show** summable  $(\lambda n. D\ (\text{real}\ n + x))$   
**proof** (rule summable-comparison-test[OF - \*], rule exI[of - 2], safe)  
**fix**  $n :: \text{nat}$  **assume**  $n \geq 2$   
**show** norm  $(D\ (\text{real}\ n + x)) \leq 1 / (12 * (x + \text{real}\ n)^2) - 1 / (12 * (x + \text{real}\ (\text{Suc}\ n))^2)$   
**using** stirling-trapezium[of real n + x]  $x$  **by** (auto simp: algebra-simps)  
**qed**  
**qed**

**private lemma**  $p$ -LIMSEQ:

**assumes**  $x: x > 0$   
**shows**  $(\lambda n. p\ n\ x) \longrightarrow P\ x$   
**proof** –  
**from**  $D$ -summable[OF  $x$ ] **have**  $(\lambda n. D\ (\text{real}\ n + x))$  sums  $P\ x$  **unfolding**  $P$ -def  
**by** (simp add: sums-iff)  
**hence**  $(\lambda n. \sum r < n. D\ (\text{real}\ r + x)) \longrightarrow P\ x$  **by** (simp add: sums-def)  
**moreover from** eventually-gt-at-top[of 1]  
**have** eventually  $(\lambda n. (\sum r < n. D\ (\text{real}\ r + x)) = p\ n\ x)$  at-top  
**by** (auto elim!: eventually-mono simp: p-def)  
**ultimately show** ?thesis **by** (rule Lim-transform-eventually [rotated])  
**qed**

This gives us an expansion of the Digamma function:

**lemma**  $Digamma$ -approx:

**assumes**  $x: (x :: \text{real}) > 0$   
**shows**  $Digamma\ x = \ln\ x - 1 / (2 * x) - P\ x$   
**proof** –  
**have** eventually  $(\lambda n. \ln\ (\text{inverse}\ (1 + x / \text{real}\ n)) + \ln\ x - 1 / (2 * x) - p\ n\ x = \ln\ (\text{real}\ n) - S'\ n\ x - 1 / (2 * x))$  at-top  
**using** eventually-gt-at-top[of 1::nat]  
**proof** eventually-elim  
**fix**  $n :: \text{nat}$  **assume**  $n: n > 1$   
**have**  $\ln\ (\text{real}\ n) - S'\ n\ x = \ln\ ((\text{real}\ n) / (\text{real}\ n + x)) + \ln\ x - p\ n\ x$   
**using** assms  $n$  **unfolding**  $S'$ -approx **by** (subst ln-div) (auto simp: algebra-simps)  
**also from**  $n$  **have**  $\text{real}\ n / (\text{real}\ n + x) = \text{inverse}\ (1 + x / \text{real}\ n)$  **by** (simp add: field-simps)  
**finally show**  $\ln\ (\text{inverse}\ (1 + x / \text{real}\ n)) + \ln\ x - 1 / (2 * x) - p\ n\ x =$

$\ln (\text{real } n) - S' n x - 1/(2*x)$  **by** *simp*

**qed**

**moreover have**  $(\lambda n. \ln (\text{inverse } (1 + x / \text{real } n)) + \ln x - 1/(2*x) - p n x)$   
 $\longrightarrow \ln (\text{inverse } (1 + 0)) + \ln x - 1/(2*x) - P x$

**by** (*rule tendsto-intros p-LIMSEQ x real-tendsto-divide-at-top*  
*filterlim-real-sequentially | simp*)**+**

**hence**  $(\lambda n. \ln (\text{inverse } (1 + x / \text{real } n)) + \ln x - 1/(2*x) - p n x)$   
 $\longrightarrow \ln x - 1/(2*x) - P x$  **by** *simp*

**ultimately have**  $(\lambda n. \ln (\text{real } n) - S' n x - 1 / (2 * x)) \longrightarrow \ln x - 1/(2*x)$   
 $- P x$

**by** (*rule Lim-transform-eventually*)

**moreover from**  $x$  **have**  $(\lambda n. \ln (\text{real } n) - S' n x - 1 / (2 * x)) \longrightarrow$   
*Digamma x*

**by** (*intro S'-LIMSEQ-Digamma simp-all*)

**ultimately show**  $\text{Digamma } x = \ln x - 1 / (2 * x) - P x$

**by** (*rule LIMSEQ-unique [rotated]*)

**qed**

Next, we derive some bounds on  $P$ :

**private lemma** *p-ge-0*:  $x > 0 \implies p n x \geq 0$   
**using** *stirling-trapezium*[of  $\text{real } n + x$  for  $n$ ]  
**by** (*auto simp add: p-def intro!: sum-nonneg*)

**private lemma** *P-ge-0*:  $x > 0 \implies P x \geq 0$   
**by** (*rule tendsto-lowerbound[OF p-LIMSEQ]*)  
*(insert p-ge-0*[of  $x$ ], *simp-all*)

**private lemma** *p-upper-bound*:

**assumes**  $x > 0 n > 0$   
**shows**  $p n x \leq 1/(12*x^2)$

**proof** –

**from** *assms* **have**  $p n x = (\sum r < n. D (\text{real } r + x))$   
**by** (*simp add: p-def*)

**also have**  $\dots \leq (\sum r < n. 1/(12*(\text{real } r + x)^2) - 1/(12 * (\text{real } (\text{Suc } r) + x)^2))$

**using** *stirling-trapezium*[of  $\text{real } r + x$  for  $r$ ] *assms*  
**by** (*intro sum-mono*) (*simp add: add-ac*)

**also have**  $\dots = 1 / (12 * x^2) - 1 / (12 * (\text{real } n + x)^2)$   
**by** (*subst sum-lessThan-telescope*) *simp*

**also have**  $\dots \leq 1 / (12 * x^2)$  **by** *simp*

**finally show** *?thesis* .

**qed**

**private lemma** *P-upper-bound*:

**assumes**  $x > 0$   
**shows**  $P x \leq 1/(12*x^2)$

**proof** (*rule tendsto-upperbound*)

**show** *eventually*  $(\lambda n. p n x \leq 1 / (12 * x^2))$  *at-top*  
**using** *eventually-gt-at-top*[of  $0$ ] *p-upper-bound*[of  $x$ ] *assms*

```

  by (auto elim!: eventually-mono)
  show  $(\lambda n. p n x) \longrightarrow P x$ 
  by (simp add: assms p-LIMSEQ)
qed auto

```

We can now use this approximation of the Digamma function to approximate *ln-Gamma* (since the former is the derivative of the latter). We therefore define the function *g* from Jameson's article, which measures the difference between *ln-Gamma* and its approximation:

```

private definition g :: real  $\Rightarrow$  real where
  g x = ln-Gamma x - (x - 1/2) * ln x + x

```

```

private lemma DERIV-g:  $x > 0 \implies (g \text{ has-field-derivative } -P x) (at x)$ 
  unfolding g-def [abs-def] using Digamma-approx[of x]
  by (auto intro!: derivative-eq-intros simp: field-simps)

```

```

private lemma isCont-P:

```

```

  assumes  $x > 0$ 
  shows isCont P x

```

```

proof -

```

```

  define D' :: real  $\Rightarrow$  real

```

```

  where  $D' = (\lambda x. -1 / (2 * x^2 * (x+1)^2))$ 

```

```

  have DERIV-D:  $(D \text{ has-field-derivative } D' x) (at x)$  if  $x > 0$  for x

```

```

  unfolding D-def [abs-def] D'-def T-def

```

```

  by (insert that, (rule derivative-eq-intros refl | simp)+)

```

```

  (simp add: power2-eq-square divide-simps, (simp add: algebra-simps)?)

```

```

  note this [THEN DERIV-chain2, derivative-intros]

```

```

  have  $(P \text{ has-field-derivative } (\sum n. D' (real n + x))) (at x)$ 

```

```

  unfolding P-def [abs-def]

```

```

  proof (rule has-field-derivative-series')

```

```

    show convex  $\{x/2<..\}$  by simp

```

```

  next

```

```

    fix n :: nat and y :: real assume  $y: y \in \{x/2<..\}$ 

```

```

    with assms have  $y > 0$  by simp

```

```

    thus  $((\lambda a. D (real n + a)) \text{ has-real-derivative } D' (real n + y)) (at y \text{ within } \{x/2<..\})$ 

```

```

    by (auto intro!: derivative-eq-intros)

```

```

  next

```

```

    from assms D-summable[of x] show summable  $(\lambda n. D (real n + x))$  by simp

```

```

  next

```

```

    show uniformly-convergent-on  $\{x/2<..\}$   $(\lambda n x. \sum i<n. D' (real i + x))$ 

```

```

  proof (rule weierstrass-m-test')

```

```

    fix n :: nat and y :: real

```

```

    assume  $y: y \in \{x/2<..\}$ 

```

```

    with assms have  $y > 0$  by auto

```

```

    have  $\text{norm } (D' (real n + y)) = (1 / (2 * (y + real n)^2)) * (1 / (y + real (Suc n))^2)$ 

```

```

    by (simp add: D'-def add-ac)

```



**also from**  $y$  *assms* **have**  $\dots \leq (1 / (2 * (x/2)^2)) * (1 / (\text{real } (\text{Suc } n))^2)$   
**by** (*intro mult-mono divide-left-mono power-mono simp-all*)  
**also have**  $1 / (\text{real } (\text{Suc } n))^2 = \text{inverse } ((\text{real } (\text{Suc } n))^2)$  **by** (*simp add: field-simps*)  
**finally show**  $\text{norm } (D' (\text{real } n + y)) \leq (1 / (2 * (x/2)^2)) * \text{inverse } ((\text{real } (\text{Suc } n))^2)$ .  
**next**  
**show** *summable*  $(\lambda n. (1 / (2 * (x/2)^2)) * \text{inverse } ((\text{real } (\text{Suc } n))^2))$   
**by** (*subst summable-Suc-iff, intro summable-mult inverse-power-summable simp-all*)  
**qed**  
**qed** (*insert assms, simp-all add: interior-open*)  
**thus** *?thesis* **by** (*rule DERIV-isCont*)  
**qed**

**private lemma** *P-continuous-on* [*THEN continuous-on-subset*]: *continuous-on*  $\{0 < ..\}$   
 $P$   
**by** (*intro continuous-at-imp-continuous-on ballI isCont-P auto*)

**private lemma** *P-integrable*:  
**assumes**  $a: a > 0$   
**shows** *P integrable-on*  $\{a.. \}$   
**proof** –  
**define**  $f$  **where**  $f = (\lambda n x. \text{if } x \in \{a.. \text{real } n\} \text{ then } P x \text{ else } 0)$   
**show** *P integrable-on*  $\{a.. \}$   
**proof** (*rule dominated-convergence*)  
**fix**  $n :: \text{nat}$   
**from**  $a$  **have** *P integrable-on*  $\{a.. \text{real } n\}$   
**by** (*intro integrable-continuous-real P-continuous-on auto*)  
**hence**  $f n$  *integrable-on*  $\{a.. \text{real } n\}$   
**by** (*rule integrable-eq [rotated]*) (*simp add: f-def*)  
**thus**  $f n$  *integrable-on*  $\{a.. \}$   
**by** (*rule integrable-on-superset [rotated 2]*) (*auto simp: f-def*)  
**next**  
**fix**  $n :: \text{nat}$   
**show**  $\forall x \in \{a.. \}. \text{norm } (f n x) \leq \text{of-real } (1/12) * (1 / x^2)$   
**using**  $a$  *P-ge-0 P-upper-bound* **by** (*auto simp: f-def*)  
**next**  
**show**  $(\lambda x :: \text{real}. \text{of-real } (1/12) * (1 / x^2))$  *integrable-on*  $\{a.. \}$   
**using** *has-integral-inverse-power-to-inf [of 2 a] a*  
**by** (*intro integrable-on-cmult-left auto*)  
**next**  
**show**  $\forall x \in \{a.. \}. (\lambda n. f n x) \longrightarrow P x$   
**proof** *safe*  
**fix**  $x :: \text{real}$  **assume**  $x: x \geq a$   
**have** *eventually*  $(\lambda n. \text{real } n \geq x)$  *at-top*  
**using** *filterlim-real-sequentially* **by** (*simp add: filterlim-at-top*)  
**with**  $x$  **have** *eventually*  $(\lambda n. f n x = P x)$  *at-top*  
**by** (*auto elim!: eventually-mono simp: f-def*)

```

    thus ( $\lambda n. f n x$ )  $\longrightarrow$   $P x$  by (simp add: Lim-eventually)
  qed
qed
qed

```

**private definition**  $c :: \text{real}$  **where**  $c = \text{integral } \{1..\} (\lambda x. -P x) + g 1$

We can now give bounds on  $g$ :

**private lemma**  $g\text{-bounds}$ :

```

  assumes  $x: x \geq 1$ 
  shows  $g x \in \{c..c + 1/(12*x)\}$ 
proof -
  from assms have int-nonneg:  $\text{integral } \{x..\} P \geq 0$ 
  by (intro Henstock-Kurzweil-Integration.integral-nonneg  $P$ -integrable)
  (auto simp:  $P$ -ge-0)
  have int-upper-bound:  $\text{integral } \{x..\} P \leq 1/(12*x)$ 
  proof (rule has-integral-le)
    from  $x$  show ( $P$  has-integral  $\text{integral } \{x..\} P$ )  $\{x..\}$ 
    by (intro integrable-integral  $P$ -integrable) simp-all
    from  $x$  has-integral-mult-right[OF has-integral-inverse-power-to-inf[of 2  $x$ ], of
    1/12]
    show (( $\lambda x. 1/(12*x^2)$ ) has-integral ( $1/(12*x)$ ))  $\{x..\}$  by (simp add:
    field-simps)
  qed (insert  $P$ -upper-bound  $x$ , simp-all)

  note DERIV-g [THEN DERIV-chain2, derivative-intros]
  from assms have int1: (( $\lambda x. -P x$ ) has-integral ( $g x - g 1$ ))  $\{1..x\}$ 
  by (intro fundamental-theorem-of-calculus)
  (auto simp: has-field-derivative-iff-has-vector-derivative [symmetric]
  intro!: derivative-eq-intros)
  from  $x$  have int2: (( $\lambda x. -P x$ ) has-integral  $\text{integral } \{x..\} (\lambda x. -P x)$ )  $\{x..\}$ 
  by (intro integrable-integral integrable-neg  $P$ -integrable) simp-all
  from has-integral-union[OF int1 int2]  $x$ 
  have (( $\lambda x. -P x$ ) has-integral  $g x - g 1 - \text{integral } \{x..\} P$ ) ( $\{1..x\} \cup \{x..\}$ )
  by (simp add: max-def)
  also from  $x$  have  $\{1..x\} \cup \{x..\} = \{1..\}$  by auto
  finally have (( $\lambda x. -P x$ ) has-integral  $g x - g 1 - \text{integral } \{x..\} P$ )  $\{1..\}$  .
  moreover have (( $\lambda x. -P x$ ) has-integral  $\text{integral } \{1..\} (\lambda x. -P x)$ )  $\{1..\}$ 
  by (intro integrable-integral integrable-neg  $P$ -integrable) simp-all
  ultimately have  $g x - g 1 - \text{integral } \{x..\} P = \text{integral } \{1..\} (\lambda x. -P x)$ 
  by (simp add: has-integral-unique)
  hence  $g x = c + \text{integral } \{x..\} P$  by (simp add: c-def algebra-simps)
  with int-upper-bound int-nonneg show  $g x \in \{c..c + 1/(12*x)\}$  by simp
qed

```

Finally, we have bounds on  $\ln$ -Gamma, Gamma, and fact.

**private lemma**  $\ln$ -Gamma-bounds-aux:

```

 $x \geq 1 \implies \ln\text{-Gamma } x \geq c + (x - 1/2) * \ln x - x$ 
 $x \geq 1 \implies \ln\text{-Gamma } x \leq c + (x - 1/2) * \ln x - x + 1/(12*x)$ 

```

**using**  $g$ -bounds[*of x*] **by** (*simp-all add: g-def*)

**private lemma** *Gamma-bounds-aux*:  
**assumes**  $x: x \geq 1$   
**shows**  $\Gamma x \geq \exp c * x \text{ powr } (x - 1/2) / \exp x$   
 $\Gamma x \leq \exp c * x \text{ powr } (x - 1/2) / \exp x * \exp (1/(12*x))$   
**proof** –  
**have**  $\exp (\ln \Gamma x) \geq \exp (c + (x - 1/2) * \ln x - x)$   
**by** (*subst exp-le-cancel-iff, rule ln-Gamma-bounds-aux*) (*simp add: x*)  
**with**  $x$  **show**  $\Gamma x \geq \exp c * x \text{ powr } (x - 1/2) / \exp x$   
**by** (*simp add: Gamma-real-pos-exp exp-add exp-diff powr-def del: exp-le-cancel-iff*)  
**next**  
**have**  $\exp (\ln \Gamma x) \leq \exp (c + (x - 1/2) * \ln x - x + 1/(12*x))$   
**by** (*subst exp-le-cancel-iff, rule ln-Gamma-bounds-aux*) (*simp add: x*)  
**with**  $x$  **show**  $\Gamma x \leq \exp c * x \text{ powr } (x - 1/2) / \exp x * \exp (1/(12*x))$   
**by** (*simp add: Gamma-real-pos-exp exp-add exp-diff powr-def del: exp-le-cancel-iff*)  
**qed**

**private lemma** *Gamma-asympt-equiv-aux*:  
 $\Gamma \sim (\lambda x. \exp c * x \text{ powr } (x - 1/2) / \exp x)$   
**proof** (*rule asympt-equiv-sandwich*)  
**show** *eventually*  $(\lambda x. \exp c * x \text{ powr } (x - 1/2) / \exp x \leq \Gamma x)$  *at-top*  
*eventually*  $(\lambda x. \exp c * x \text{ powr } (x - 1/2) / \exp x * \exp (1/(12*x)) \geq \Gamma x)$  *at-top*  
**using** *eventually-ge-at-top*[*of 1::real*] *Gamma-bounds-aux*  
**by** (*auto elim!: eventually-mono*)  
**have**  $((\lambda x::\text{real}. \exp (1 / (12 * x))) \longrightarrow \exp 0)$  *at-top*  
**by** (*rule tendsto-intros real-tendsto-divide-at-top filterlim-tendsto-pos-mult-at-top*) +  
*(simp-all add: filterlim-ident)*  
**hence**  $(\lambda x. \exp (1 / (12 * x))) \sim (\lambda x. 1)$   
**by** (*intro asympt-equivI'*) *simp-all*  
**hence**  $(\lambda x. \exp c * x \text{ powr } (x - 1 / 2) / \exp x * 1) \sim$   
 $(\lambda x. \exp c * x \text{ powr } (x - 1 / 2) / \exp x * \exp (1 / (12 * x)))$   
**by** (*intro asympt-equiv-mult asympt-equiv-refl*) (*simp add: asympt-equiv-sym*)  
**thus**  $(\lambda x. \exp c * x \text{ powr } (x - 1 / 2) / \exp x) \sim$   
 $(\lambda x. \exp c * x \text{ powr } (x - 1 / 2) / \exp x * \exp (1 / (12 * x)))$  **by** *simp*  
**qed** *simp-all*

**private lemma** *exp-1-powr-real* [*simp*]:  $\exp (1::\text{real}) \text{ powr } x = \exp x$   
**by** (*simp add: powr-def*)

**private lemma** *fact-asympt-equiv-aux*:  
 $\text{fact} \sim (\lambda x. \exp c * \text{sqrt } (\text{real } x) * (\text{real } x / \exp 1) \text{ powr } \text{real } x)$   
**proof** –  
**have**  $\text{fact} \sim (\lambda n. \Gamma (\text{real } (\text{Suc } n)))$  **by** (*simp add: Gamma-fact*)  
**also have** *eventually*  $(\lambda n. \Gamma (\text{real } (\text{Suc } n)) = \text{real } n * \Gamma (\text{real } n))$   
*at-top*  
**using** *eventually-gt-at-top*[*of 0::nat*] *Gamma-plus1*[*of real n for n*]  
**by** (*auto elim!: eventually-mono simp: add-ac of-nat-in-nonpos-Ints-iff*)

**also have**  $(\lambda n. \text{Gamma } (\text{real } n)) \sim (\lambda n. \text{exp } c * \text{real } n \text{ powr } (\text{real } n - 1/2) / \text{exp } (\text{real } n))$   
**by**  $(\text{rule } \text{asympt-equiv-compose}'[\text{OF Gamma-asympt-equiv-aux}] \text{filterlim-real-sequentially}) +$   
**also have**  $\text{eventually } (\lambda n. \text{real } n * (\text{exp } c * \text{real } n \text{ powr } (\text{real } n - 1 / 2) / \text{exp } (\text{real } n))) =$   
 $\text{exp } c * \text{sqrt } (\text{real } n) * (\text{real } n / \text{exp } 1) \text{ powr } \text{real } n \text{ at-top}$   
**using**  $\text{eventually-gt-at-top}[\text{of } 0::\text{nat}]$   
**proof**  $\text{eventually-elim}$   
**fix**  $n :: \text{nat}$  **assume**  $n: n > 0$   
**thus**  $\text{real } n * (\text{exp } c * \text{real } n \text{ powr } (\text{real } n - 1 / 2) / \text{exp } (\text{real } n)) =$   
 $\text{exp } c * \text{sqrt } (\text{real } n) * (\text{real } n / \text{exp } 1) \text{ powr } \text{real } n$   
**by**  $(\text{subst } \text{powr-divide2 } [\text{symmetric}]) (\text{simp-all add: powr-divide powr-half-sqrt field-simps})$   
**qed**  
**finally show**  $?thesis$  **by**  $- (\text{simp-all add: asympt-equiv-mult})$   
**qed**

We still need to determine the constant term  $c$ , which we do using Wallis' product formula:

$$\prod_{n=1}^{\infty} \frac{4n^2}{4n^2 - 1} = \frac{\pi}{2}$$

**private lemma**  $\text{powr-mult-2: } (x::\text{real}) > 0 \implies x \text{ powr } (y * 2) = (x^2) \text{ powr } y$   
**by**  $(\text{subst } \text{mult.commute}, \text{subst } \text{powr-powr } [\text{symmetric}]) (\text{simp add: powr-numeral})$

**private lemma**  $\text{exp-mult-2: } \text{exp } (y * 2 :: \text{real}) = \text{exp } y * \text{exp } y$   
**by**  $(\text{subst } \text{exp-add } [\text{symmetric}]) \text{simp}$

**private lemma**  $\text{exp-c: } \text{exp } c = \text{sqrt } (2 * \pi)$

**proof**  $-$

**define**  $p$  **where**  $p = (\lambda n. \prod_{k=1..n}. (4 * \text{real } k^2) / (4 * \text{real } k^2 - 1))$

**have**  $p-0$   $[\text{simp}]: p\ 0 = 1$  **by**  $(\text{simp add: p-def})$

**have**  $p\text{-Suc: } p (\text{Suc } n) = p * (4 * \text{real } (\text{Suc } n)^2) / (4 * \text{real } (\text{Suc } n)^2 - 1)$

**for**  $n$  **unfolding**  $p\text{-def}$  **by**  $(\text{subst } \text{prod-nat-ivl-Suc}') \text{simp-all}$

**have**  $p: p = (\lambda n. 16^n * \text{fact } n^4 / (\text{fact } (2 * n))^2 / (2 * \text{real } n + 1))$

**proof**

**fix**  $n :: \text{nat}$

**have**  $p\ n = (\prod_{k=1..n}. (2 * \text{real } k)^2 / (2 * \text{real } k - 1) / (2 * \text{real } k + 1))$

**unfolding**  $p\text{-def}$  **by**  $(\text{intro } \text{prod.cong refl}) (\text{simp add: field-simps power2-eq-square})$

**also have**  $\dots = (\prod_{k=1..n}. (2 * \text{real } k)^2 / (2 * \text{real } k - 1)) / (\prod_{k=1..n}. (2 * \text{real } (\text{Suc } k) - 1))$

**by**  $(\text{simp add: prod-dividef prod.distrib add-ac})$

**also have**  $(\prod_{k=1..n}. (2 * \text{real } (\text{Suc } k) - 1)) = (\prod_{k=\text{Suc } 1.. \text{Suc } n}. (2 * \text{real } k - 1))$

**by**  $(\text{subst } \text{prod.atLeast-Suc-atMost-Suc-shift}) \text{simp-all}$

**also have**  $\dots = (\prod_{k=1..n}. (2 * \text{real } k - 1)) * (2 * \text{real } n + 1)$

**by**  $(\text{induction } n) (\text{simp-all add: prod-nat-ivl-Suc}')$

**also have**  $(\prod_{k=1..n}. (2 * \text{real } k)^2 / (2 * \text{real } k - 1)) / \dots =$

$(\prod k = 1..n. (2 * \text{real } k)^2 / (2 * \text{real } k - 1)^2) / (2 * \text{real } n + 1)$   
**unfolding** *power2-eq-square* **by** (*simp add: prod.distrib prod-dividef*)  
**also have**  $(\prod k = 1..n. (2 * \text{real } k)^2 / (2 * \text{real } k - 1)^2) =$   
 $(\prod k = 1..n. (2 * \text{real } k)^4 / ((2 * \text{real } k) * (2 * \text{real } k - 1))^2)$   
**by** (*intro prod.cong refl*)  
*(simp add: divide-simps, (simp add: field-simps power2-eq-square eval-nat-numeral))*  
**also have**  $\dots = 16^n * \text{fact } n^4 / (\prod k=1..n. (2 * \text{real } k) * (2 * \text{real } k - 1))^2$   
**by** (*simp add: prod.distrib prod-dividef fact-prod*  
*prod-power-distrib [symmetric] prod-constant*)  
**also have**  $(\prod k=1..n. (2 * \text{real } k) * (2 * \text{real } k - 1)) = \text{fact } (2 * n)$   
**by** (*induction n*) (*simp-all add: algebra-simps prod-nat-ivl-Suc'*)  
**finally show**  $p \ n = 16^n * \text{fact } n^4 / (\text{fact } (2 * n))^2 / (2 * \text{real } n + 1) .$   
**qed**

**have**  $p \sim (\lambda n. 16^n * \text{fact } n^4 / (\text{fact } (2 * n))^2 / (2 * \text{real } n + 1))$   
**by** (*simp add: p*)  
**also have**  $\dots \sim (\lambda n. 16^n * (\exp c * \text{sqrt } (\text{real } n)) * (\text{real } n / \exp 1) \text{ powr real } n)^4 /$   
 $(\exp c * \text{sqrt } (\text{real } (2 * n))) * (\text{real } (2 * n) / \exp 1) \text{ powr real } (2 * n))^2 /$   
 $(2 * \text{real } n + 1)$  (*is - ~ ?f*)  
**by** (*intro asymp-equiv-mult asymp-equiv-divide asymp-equiv-refl mult-nat-left-at-top*  
*fact-asymp-equiv-aux asymp-equiv-power asymp-equiv-compose [OF*  
*fact-asymp-equiv-aux]*  
*simp-all*)  
**also have** *eventually*  $(\lambda n. \dots \ n = \exp c^2 / (4 + 2/n))$  *at-top*  
**using** *eventually-gt-at-top*[*of 0::nat*]  
**proof** *eventually-elim*  
**fix**  $n :: \text{nat}$  **assume**  $n > 0$   
**have** [*simp*]:  $16^n = 4^n * (4^n :: \text{real})$  **by** (*simp add: power-mult-distrib*  
[*symmetric*])  
**from**  $n$  **have**  $?f \ n = \exp c^2 * (n / (2 * (2 * n + 1)))$   
**by** (*simp add: power-mult-distrib divide-simps powr-mult powr-divide real-sqrt-power-even*)  
*(simp add: field-simps power2-eq-square eval-nat-numeral powr-mult-2*  
*exp-mult-2 powr-realpow)*  
**also from**  $n$  **have**  $\dots = \exp c^2 / (4 + 2/n)$  **by** (*simp add: field-simps*)  
**finally show**  $?f \ n = \dots .$   
**qed**

**also have**  $(\lambda x. 4 + 2 / \text{real } x) \sim (\lambda x. 4)$   
**by** (*subst asymp-equiv-add-right*) *auto*  
**finally have**  $p \longrightarrow \exp c^2 / 4$   
**by** (*rule asymp-equivD-const*) (*simp-all add: asymp-equiv-divide*)  
**moreover have**  $p \longrightarrow \pi / 2$  **unfolding** *p-def* **by** (*rule wallis*)  
**ultimately have**  $\exp c^2 / 4 = \pi / 2$  **by** (*rule LIMSEQ-unique*)  
**hence**  $2 * \pi = \exp c^2$  **by** *simp*  
**also have**  $\text{sqrt } (\exp c^2) = \exp c$  **by** *simp*  
**finally show**  $\exp c = \text{sqrt } (2 * \pi) ..$   
**qed**

**private lemma**  $c: c = \ln (2*\pi) / 2$   
**proof** –  
  **note**  $\text{exp-c}$  [symmetric]  
  **also have**  $\ln (\text{exp } c) = c$  **by**  $\text{simp}$   
  **finally show**  $?thesis$  **by** ( $\text{simp add: ln-sqrt}$ )  
**qed**

This gives us the final bounds:

**theorem**  $\text{Gamma-bounds}$ :  
  **assumes**  $x \geq 1$   
  **shows**  $\text{Gamma } x \geq \text{sqrt } (2*\pi/x) * (x / \text{exp } 1) \text{ powr } x$  (**is**  $?th1$ )  
    $\text{Gamma } x \leq \text{sqrt } (2*\pi/x) * (x / \text{exp } 1) \text{ powr } x * \text{exp } (1 / (12 * x))$  (**is**  $?th2$ )  
**proof** –  
  **from**  $\text{assms}$  **have**  $\text{exp } c * x \text{ powr } (x - 1/2) / \text{exp } x = \text{sqrt } (2*\pi/x) * (x / \text{exp } 1) \text{ powr } x$   
  **by** ( $\text{subst powr-divide2}$  [symmetric])  
  ( $\text{simp add: exp-c real-sqrt-divide powr-divide powr-half-sqrt field-simps}$ )  
  **with**  $\text{Gamma-bounds-aux}[OF \text{ assms}]$  **show**  $?th1 ?th2$  **by**  $\text{simp-all}$   
**qed**

**theorem**  $\text{ln-Gamma-bounds}$ :  
  **assumes**  $x \geq 1$   
  **shows**  $\text{ln-Gamma } x \geq \ln (2*\pi/x) / 2 + x * \ln x - x$  (**is**  $?th1$ )  
    $\text{ln-Gamma } x \leq \ln (2*\pi/x) / 2 + x * \ln x - x + 1/(12*x)$  (**is**  $?th2$ )  
**proof** –  
  **from**  $\text{ln-Gamma-bounds-aux}[OF \text{ assms}]$   $\text{assms}$  **show**  $?th1 ?th2$   
  **by** ( $\text{simp-all add: c field-simps ln-div}$ )  
  **from**  $\text{assms}$  **have**  $\text{exp } c * x \text{ powr } (x - 1/2) / \text{exp } x = \text{sqrt } (2*\pi/x) * (x / \text{exp } 1) \text{ powr } x$   
  **by** ( $\text{subst powr-divide2}$  [symmetric])  
  ( $\text{simp add: exp-c real-sqrt-divide powr-divide powr-half-sqrt field-simps}$ )  
**qed**

**theorem**  $\text{fact-bounds}$ :  
  **assumes**  $n > 0$   
  **shows**  $(\text{fact } n :: \text{real}) \geq \text{sqrt } (2*\pi*n) * (n / \text{exp } 1) ^ n$  (**is**  $?th1$ )  
    $(\text{fact } n :: \text{real}) \leq \text{sqrt } (2*\pi*n) * (n / \text{exp } 1) ^ n * \text{exp } (1 / (12 * n))$  (**is**  $?th2$ )  
**proof** –  
  **from**  $\text{assms}$  **have**  $n: \text{real } n \geq 1$  **by**  $\text{simp}$   
  **from**  $\text{assms}$   $\text{Gamma-plus1}[of \text{ real } n]$   
  **have**  $\text{real } n * \text{Gamma } (\text{real } n) = \text{Gamma } (\text{real } (\text{Suc } n))$   
  **by** ( $\text{simp add: of-nat-in-nonpos-Ints-iff add-ac}$ )  
  **also have**  $\text{Gamma } (\text{real } (\text{Suc } n)) = \text{fact } n$  **by** ( $\text{subst Gamma-fact}$  [symmetric])  
 $\text{simp}$   
  **finally have**  $*$ :  $\text{fact } n = \text{real } n * \text{Gamma } (\text{real } n)$  **by**  $\text{simp}$   
  
  **have**  $2*\pi/n = 2*\pi*n / n^2$  **by** ( $\text{simp add: power2-eq-square}$ )

also have  $\text{sqrt } \dots = \text{sqrt } (2*\text{pi}*n) / n$  by (subst real-sqrt-divide) simp-all  
 also have  $\text{real } n * \dots = \text{sqrt } (2*\text{pi}*n)$  by simp  
 finally have \*\*:  $\text{real } n * \text{sqrt } (2*\text{pi}/\text{real } n) = \text{sqrt } (2*\text{pi}*\text{real } n)$  .

note \*

also note  $\text{Gamma-bounds}(2)[OF\ n]$

also have  $\text{real } n * (\text{sqrt } (2 * \text{pi} / \text{real } n) * (\text{real } n / \text{exp } 1) \text{ powr } \text{real } n * \text{exp } (1 / (12 * \text{real } n))) =$   
 $(\text{real } n * \text{sqrt } (2*\text{pi}/n)) * (n / \text{exp } 1) \text{ powr } n * \text{exp } (1 / (12 * n))$

by (simp add: algebra-simps)

also from  $n$  have  $(\text{real } n / \text{exp } 1) \text{ powr } \text{real } n = (\text{real } n / \text{exp } 1) ^ n$

by (subst powr-realpow) simp-all

also note \*\*

finally show ?th2 by – (insert assms, simp-all)

have  $\text{sqrt } (2*\text{pi}*n) * (n / \text{exp } 1) \text{ powr } n = n * (\text{sqrt } (2*\text{pi}/n) * (n / \text{exp } 1) \text{ powr } n)$

by (subst \*\* [symmetric]) (simp add: field-simps)

also from assms have  $\dots \leq \text{real } n * \text{Gamma } (\text{real } n)$

by (intro mult-left-mono  $\text{Gamma-bounds}(1)$ ) simp-all

also from  $n$  have  $(\text{real } n / \text{exp } 1) \text{ powr } \text{real } n = (\text{real } n / \text{exp } 1) ^ n$

by (subst powr-realpow) simp-all

also note \* [symmetric]

finally show ?th1 .

qed

theorem  $\text{ln-fact-bounds}$ :

assumes  $n > 0$

shows  $\text{ln } (\text{fact } n :: \text{real}) \geq \text{ln } (2*\text{pi}*n)/2 + n * \text{ln } n - n$  (is ?th1)

$\text{ln } (\text{fact } n :: \text{real}) \leq \text{ln } (2*\text{pi}*n)/2 + n * \text{ln } n - n + 1/(12*\text{real } n)$  (is ?th2)

proof –

have  $\text{ln } (\text{fact } n :: \text{real}) \geq \text{ln } (\text{sqrt } (2*\text{pi}*\text{real } n) * (\text{real } n / \text{exp } 1) ^ n)$

using  $\text{fact-bounds}(1)[OF\ \text{assms}]$  assms by (subst ln-le-cancel-iff) auto

also from assms have  $\text{ln } (\text{sqrt } (2*\text{pi}*\text{real } n) * (\text{real } n / \text{exp } 1) ^ n) = \text{ln } (2*\text{pi}*n)/2 + n * \text{ln } n - n$

by (simp add: ln-mult ln-div ln-realpow ln-sqrt field-simps)

finally show ?th1 .

next

have  $\text{ln } (\text{fact } n :: \text{real}) \leq \text{ln } (\text{sqrt } (2*\text{pi}*\text{real } n) * (\text{real } n / \text{exp } 1) ^ n * \text{exp } (1/(12*\text{real } n)))$

using  $\text{fact-bounds}(2)[OF\ \text{assms}]$  assms by (subst ln-le-cancel-iff) auto

also from assms have  $\dots = \text{ln } (2*\text{pi}*n)/2 + n * \text{ln } n - n + 1/(12*\text{real } n)$

by (simp add: ln-mult ln-div ln-realpow ln-sqrt field-simps)

finally show ?th2 .

qed

theorem  $\text{Gamma-asympt-equiv}$ :

$\text{Gamma} \sim (\lambda x. \text{sqrt } (2*\text{pi}/x) * (x / \text{exp } 1) \text{ powr } x :: \text{real})$

**proof** –  
**note** *Gamma-asymp-equiv-aux*  
**also have** *eventually*  $(\lambda x. \exp c * x \text{ powr } (x - 1/2) / \exp x = \text{sqrt } (2 * \pi / x) * (x / \exp 1) \text{ powr } x)$  *at-top*  
**using** *eventually-gt-at-top*[of  $0 :: \text{real}$ ]  
**proof** *eventually-elim*  
**fix**  $x :: \text{real}$  **assume**  $x > 0$   
**thus**  $\exp c * x \text{ powr } (x - 1/2) / \exp x = \text{sqrt } (2 * \pi / x) * (x / \exp 1) \text{ powr } x$   
**by** (*subst powr-divide2* [*symmetric*])  
*(simp add: exp-c powr-half-sqrt powr-divide field-simps real-sqrt-divide)*  
**qed**  
**finally show** *?thesis* .  
**qed**

**theorem** *fact-asymp-equiv*:  
 $\text{fact} \sim (\lambda n. \text{sqrt } (2 * \pi * n) * (n / \exp 1) ^ n :: \text{real})$   
**proof** –  
**note** *fact-asymp-equiv-aux*  
**also have** *eventually*  $(\lambda n. \exp c * \text{sqrt } (\text{real } n) = \text{sqrt } (2 * \pi * \text{real } n))$  *at-top*  
**using** *eventually-gt-at-top*[of  $0 :: \text{nat}$ ] **by** *eventually-elim* (*simp add: exp-c real-sqrt-mult*)  
**also have** *eventually*  $(\lambda n. (n / \exp 1) \text{ powr } n = (n / \exp 1) ^ n)$  *at-top*  
**using** *eventually-gt-at-top*[of  $0 :: \text{nat}$ ] **by** *eventually-elim* (*simp add: powr-realpow*)  
**finally show** *?thesis* .  
**qed**

**end**

**end**

## References

- [1] G. J. O. Jameson. A simple proof of Stirling’s formula for the Gamma function. *The Mathematical Gazette*, 99:68–74, 3 2015.